APPENDIX S

DETERMINATION OF FLI, FGI, AND FLR PARAMETERS

RES 17-06

M E M O R A N D U M

TO: Jayantha Obeysekera, Director, Office of Modeling

THROUGH: Luis Cadavid, Chief Hydrologic Modeler, Office of Modeling

FROM: Jenifer Barnes, Staff Hydrologic Modeler, Office of Modeling
Michelle M. Irizarry, Staff Hydrologic Modeler, Office of Modeling

DATE: August 6, 2004

SUBJECT: Determination of the FLR, FGI, and FLI Parameters for the SFWMM v5.4

This document describes the procedure used to determine the FLR, FGI and FLI static parameters for the South Florida Water Management Model (SFWMM) v5.4. Each 2-mile by 2-mile cell in the Lower East Coast (LEC) of the model domain is assigned a value for these parameters which are ratios that represent reductions to be applied to the irrigation demands calculated by the ET-Recharge program. FLR is the fraction of landscape irrigation from wastewater reuse. FGI is the fraction of golf course irrigation from wastewater reuse. FLI is the fraction of landscape irrigation receiving water from public water supply (PWS) wells. These fractions reduce the local irrigation demands calculated by the ET-Recharge (AFSIRS) program for the grid cells in the LEC.

For each LEC grid cell there is a calculated irrigation demand that results from running the ET-Recharge (AFSIRS) program (http://www.sfwmd.gov/org/pld/hsm/pubs/memo/et-recharge_2000base_memo.pdf). The ET-Recharge (AFSIRS) program calculates field-scale irrigation demands for polygons with unique land use codes, soil type, hydraulic conductivity, cell location within the SFWMM grid, etc. These polygon-scale irrigation demands are aggregated into 2-mile by 2-mile grid cells and grouped into six major water use types: landscape, golf course (urban), nursery, low volume overhead and other agriculture. For each LEC grid cell, the program outputs daily irrigation requirements by water use type, actual evapotranspiration components (e.g. total, unsaturated and unsaturated for irrigated areas) and acreage of irrigation by water use type.

A portion of the landscape irrigation demands in a given grid cell may be met by wastewater reuse. This fraction is accounted for in the FLR parameter. Public water supply can also be used to meet landscape irrigation demands and this fraction is accounted for in the FLI parameter. In the SFWMM it is assumed that the volume of PWS water contributing towards meeting landscape irrigation demands is already accounted for in the PWS pumpage but for simplification there is no direct connection between supply and demand. For each grid cell, the complement of FLI and FLR (i.e. 1 –

(FLI + FLR)) is the fraction of the landscape irrigation demands to be met (removed) from the groundwater.

A portion of the golf course irrigation demands in a given grid cell may be met by wastewater reuse. This fraction is accounted for in the FGI parameter. For each grid cell, the complement of FGI (i.e. 1 – FGI) is the fraction of the golf course irrigation demands to be met (removed) from the groundwater.

Only the portion of urban and agricultural irrigation demands to be met from the groundwater can be cutback as a result of restrictions triggered by stages at certain LEC trigger wells or when Lake Okeechobee is under supply-side management. PWS pumpage is cut back under these conditions, however FLI remains unaffected.

Geographic data describing areas using public water supply for irrigation is not available on a regional scale, therefore, a process to approximate these values has been developed. This memorandum describes the process for estimating FLR, FGI and FLI for each LEC grid cell. For simplification, these fractions are derived independent from AFSIRS based on ratios of irrigated polygon areas instead of ratios of irrigation requirements for all relevant polygons within a grid cell.

FLR and FGI

The procedure for calculating the FLR (fraction of landscape irrigation within a grid cell receiving water from wastewater reuse) and FGI (fraction of golf course irrigation within a grid cell receiving water from wastewater reuse) parameters identifies specific polygons in the land use data that are known to be using wastewater reuse for irrigation purposes. Those polygons with wastewater reuse are identified with the letter “R” placed after the FLUCCS code in an item called “Reuse”.

For each SFWMM grid cell, the fraction of landscape irrigation from reuse (FLR) is calculated by:

$$FLR = \frac{\sum_{i=1}^{N_R} polyarea \times \%pervious}{\sum_{j=1}^{M} polyarea \times \%pervious}$$

where: N_R = *# of polygons with landscape irrigation reuse code within grid cell*
M = *total # of polygons with landscape irrigation within grid cell*

For each SFWMM grid cell, the fraction of golf course irrigation from reuse (FGI) is calculated by:

$$FGI = \frac{\sum_{i=1}^{N_G} polyarea \times \%pervious}{\sum_{j=1}^{M} polyarea \times \%pervious}$$

where: N_G = *# of polygons with golf course irrigation reuse code within grid cell*
M = *total # of polygons with golf course irrigation within grid cell*

The programs used to calculate the SFWMM values are included in the appendices. First an Arc Macro Language (AML) is run in Arc/INFO to process the land use data (Appendix A) then a Fortran program is run to calculate areas and assign them to SFWMM grid cells (Appendix B). Appendix C is the lookup table that the program in Appendix B requires.

Figures 1 through 6 show results for FLR and FGI for the SFWMM v5.4 2000, 2010 and 2050 bases.



Figure 1. 2000 FLR Values



Figure 2. 2010 FLR Values



Figure 3. 2050 FLR Values



Figure 4. 2000 FGI Values



Figure 5. 2010 FGI Values



Figure 6. 2050 FGI Values

FLI

FLI is defined as the fraction of landscape irrigation within a grid cell receiving water from public water supply wells. The process developed to approximate these values utilizes land use data and permit boundaries. The SFWMD maintains a database of active landscape permits which define the boundaries of areas supplying their own irrigation water (self-supplied irrigation). The land use polygons that fall within these permit boundaries are identified by landscape permit number and omitted from the FLI calculation. This dataset is then merged with the public water utility service area boundaries. The resulting polygons are identified with a utility permit number and included in the FLI calculation. Any polygons that are already identified with the FLR or FGI parameters are also omitted from the calculations.

An FLI value is assigned to the remaining polygons, $FLI_{poly}(LU)$, and it depends on the land use type (see Table 1) and the location relative to the saltwater interface. A value is assigned to each type of land use that is included in the irrigation demand calculations. The area is then totaled for each grid cell and an input dataset is created for the SFWMM.

For each SFWMM grid cell, the fraction of landscape irrigation from PWS (FLI) is calculated by:

$$FLI = \frac{\sum_{i=1}^{M} FLI_{poly}(LU) * polyarea \times \%pervious}{\sum_{j=1}^{M} polyarea \times \%pervious}$$

where: $M =$ *total # of polygons with landscape irrigation within grid cell*

As a post-processing step, each grid cell that falls within a service area boundary and is the easternmost urban grid cell is assigned an FLI value of one. This means that no landscape irrigation demands will be taken from the groundwater system in these cells, in order to reduce the threat of salt water intrusion.

The AML used to process the land use data is included as Appendix A, which is a script used in Arc/INFO. First, the overall land use dataset is reduced to the urban areas east of the conservation areas. An item called “irrigated” is added to identify only those land use types that a landscape or golf course irrigation demand is calculated for. If a land use is identified as a reuse polygon but no demands are calculated in the ET-Recharge (AFSIRS) program it is reverted back to a non-reuse polygon (the “R” is removed from its FLUCCS code). FGI polygons are treated differently in that their reuse designation remains but they are not designated as irrigated in this process. All reuse polygons, non-irrigated and landscape irrigation permitted lands are then cleared of their utility permit numbers. This utility permit number is used as an identifier in the output file; these are the areas that are irrigated from public water supply. A text file is produced and used as

input to the program that calculates the FLR, FGI and FLI parameters. The same file is also used as input to the ET-Recharge (AFSIRS) program.

The program used to calculate FLR, FGI and FLI also calculates the acreages used for other types of irrigation such as nursery and agriculture. It is included here as Appendix B. The FLI values are defined in a lookup table so that it is easy to change them when running the program.

FLUCCS Code	Description	FLI value
120	Medium Density Residential	0.6
121	Medium Density Single Family	0.6
130	High Density Residential	0.6
131	High Density Single Family	0.6
133	High Density Low Rise (2 stories or less)	0.7
134	High Density High Rise (3 stories or more)	0.7
135	High Density Mixed Units	0.6
140	Commercial and Services	0.8
141	Retail Sales and Services	0.8
1411	Shopping Centers	0.8
142	Wholesale Sales and Services	0.8
143	Professional Services	0.8
144	Cultural and Entertainment	0.8
145	Tourist Services	0.8
147	Mixed Commercial and Services	0.8
148	Cemeteries	0.8
170	Institutional	0.8
171	Educational Facilities	0.8
172	Religious	0.8
173	Military	0.8
174	Medical and Health Care	0.8
175	Governmental	0.8
176	Correctional	0.8
180	Recreational	0.8
183	Race Tracks	0.8
185	Parks and Zoos	0.8
186	Community Recreational Facilities	0.8
187	Stadiums	0.8
188	Historical Sites	0.8
189	Other Recreational	0.8
253	Kennels	0.8
811	Airports	0.8
813	Bus and Truck Terminals	0.8
833	Water Supply Plants	0.8

Table 1. Landscape irrigation types and their FLI values.



Figure 7. 1988 FLI Values



Figure 8. 2000 FLI Values



Figure 9. 2010 FLI Values



Figure 10. 2050 FLI Values

Appendix A. AML to process land use for FLR, FGI, and FLR parameters (2000) and ET-Recharge program.

```
clip landuse_2000 lec_v50 lec_lu2000
identity lec_lu2000 lsc_permit lec_2000lsc
identity lec_2000lsc service_areas lec_2000fli
additem lec_2000fli.pat lec_2000fli.pat irrigated 8 8 c
kill lec_2000lsc all
kill lec_lu2000 all
ae
ec lec_2000fli
ef labels
sel for reuse = '120' or reuse = '121' or reuse = '123' or reuse = '130' or reuse = '120R' or
reuse = '121R' or reuse = '123R' or reuse = '130R'
calc irrigated = 'y'
sel for reuse = '131' or reuse = '133' or reuse = '134' or reuse = '135' or reuse = '131R' or
reuse = '133R' or reuse = '134R' or reuse = '135R'
calc irrigated = 'y'
sel for reuse = '1009' or reuse = '1411' or reuse = '140' or reuse = '141' or reuse = '1009R'
or reuse = '1411R' or reuse = '140R' or reuse = '141R'
calc irrigated = 'y'
sel for reuse = '142' or reuse  = '143' or reuse = '813' or reuse = '253' or reuse = '142R' or
reuse  = '143R' or reuse = '813R' or reuse = '253R'
calc irrigated = 'y'
sel for reuse = '144' or reuse = '183' or reuse = '147' or reuse = '145' or reuse = '144R' or
reuse = '183R' or reuse = '147R' or reuse = '145R'
calc irrigated = 'y'
sel for reuse = '170' or reuse = '171' or reuse = '172' or reuse = '173' or reuse = '170R' or
reuse = '171R' or reuse = '172R' or reuse = '173R'
calc irrigated = 'y'
sel for reuse = '174' or reuse = '175' or reuse = '176' or reuse = '811' or reuse = '174R' or
reuse = '175R' or reuse = '176R' or reuse = '811R'
calc irrigated = 'y'
sel for reuse = '833' or reuse = '180' or reuse = '186' or reuse = '187' or reuse = '833R' or
reuse = '180R' or reuse = '186R' or reuse = '187R'
calc irrigated = 'y'
sel for reuse = '189' or reuse = '185' or reuse = '188' or reuse = '148' or reuse = '189R' or
reuse = '185R' or reuse = '188R' or reuse = '148R' or reuse = '182R'
calc irrigated = 'y'
save
sel for reuse cn 'R' and irrigated = ''
calc reuse = lu2000
sel for reuse = '182R'
calc irrigated = ''
save
sel for reuse cn 'R'
calc permit_no = ''
```

```
calc permit_fut = ''
sel for lsc_perm ne ''
calc permit_no = ''
calc permit_fut = ''
sel for irrigated = ''
calc permit_no = ''
calc permit_fut = ''
save
quit
identity lec-etsoil83 lec_2000fli sfwmm1
identity sfwmm1 sfwmm_gr_83 sfwmm2
kill sfwmm1 all
tables
select sfwmm2.pat
UNLOAD lu_soil_2000_fli.txt area,reuse,rowco,afsirs-id,ib,vt-
perm,etbasin#,salt,salt,permit_no COLUMNAR init
q stop
kill sfwmm2 all
```

Appendix B. FORTRAN program used to create static grid values (create_statdta_all.f) for ET-Recharge, FLR, FGI, and FLI.

```
      program create
* Declarations
      character alandsoil*9
      character*9 cfname
      character*9 lu
      character*12 soilfile
      character*3 cuttype
      character*10 pwsprm
      integer row,col
      integer isoil,ib,perm,rbasin,salt
      integer jcount,iflag
      real polyarea,loadet,frac
      integer loadcf,xeriscape
      real acres,glf_reuse,lan_reuse,lan_pws,flr,fgi,fli
      real flipoly, flilu
      integer statr,statc1,statc2,ist,ord
      integer rowlec,minco,maxco
      integer code_low,code_up

      parameter(dim1=1,dim2=10,dim3=500,ncol=41,nrow=65,ncrop=6)


      dimension isoil(dim1),loadcf(dim3,8),loadet(dim3,5),
     +          cfname(dim3),
     +          rowlec(nrow),minco(nrow),maxco(nrow),
     +          cuttype(ncrop),ord(ncrop),acres(nrow,ncol,ncrop),
     +          statr(184),statc1(184),statc2(184),
     +          glf_reuse(nrow,ncol),lan_reuse(nrow,ncol),
     +          lan_pws(nrow,ncol),
     +          flr(nrow,ncol), fgi(nrow,ncol), fli(nrow,ncol),
     +          lu(dim3),flilu(dim3)

* Defining use type
      cuttype(1) = 'NUR'
      cuttype(2) = 'LSC'
      cuttype(3) = 'AOV'
      cuttype(4) = 'ALV'
      cuttype(5) = 'AOT'
      cuttype(6) = 'GLF'

* Defining order in which use types will be output
      ord(1) = 2
```

```
      ord(2) = 1
      ord(3) = 6
      ord(4) = 4
      ord(5) = 3
      ord(6) = 5

* Initializing all acreages to zero
      do 101 i=1,nrow
        do 102 j=1,ncol
          do 103 k=1,ncrop
            acres(i,j,k)=0.0
103         continue
102       continue
101     continue


* Opening/reading file with row, mincol, maxcol for statdta
      open(10,file='statcells.dat',status='old')
      do 104 i=1,184
        read(10,1001) statr(i),statc1(i), statc2(i)
1001  format(i2,1x,i2,1x,i2)
104   continue
      close(10)

* Opening file with name of file with polygon info
      open(20,file='et-sfwmm.par',status='old')
      read(20,1002)soilfile
      write(*,*) soilfile
1002  format(a12)
      close(20)

* Opening file with polygon info
      open(30,file=soilfile,status='old')
* Skipping first line
      read(30,*)

* Opening file with LU code description
      open(40,file='et-newcodes.txt',status="old")
* Skipping first line
      read(40,*)

* Opening file with flipoly definition
      open(45,file='flipoly.dat',status='old')
* Skipping first line
      read(45,*)

* Initialize counter for number of lines in file
```

```
      jline=0
* Opening/reading file with active LEC cells
      open(50,file='leccells.dat',status='old')
      do 105 il=1,nrow
        read(50,1003,end=106) rowlec(il),minco(il),maxco(il)
        jline=jline+1
1003   format(6x,i2,6x,i2,6x,i2)
105   continue
106   continue
      close(50)

* Read LU codes and description from file
      do 107 jjj=1,dim3
        read(40,1004,end=108) cfname(jjj),
     +   loadcf(jjj,1),(loadet(jjj,jmon),jmon=1,4),
     +   (loadcf(jjj,imon),imon=2,8),loadet(jjj,5)
1004  format(a9,i2,6x,4(F5.2),5(3x,i2),i5,i4,f6.2)
107   continue
108   continue
      close(40)

* Read flipoly for each LU
      do 125 jjj=1,dim3
        read(45,1010,end=126) lu(jjj),flilu(jjj)
1010  format(a9,f5.3)
125   continue
126   continue
      close(45)

* Read polygon description from file
      do 109 k=1,13400000
        read(30,1005,end=110) polyarea,alandsoil,row,col,isoil
     +   ,ib,perm,rbasin,salt,xeriscape,pwsprm
* Change empty LU codes to open water
        if(alandsoil .eq. '    ') alandsoil='521 '
1005  format(f18.5,a5,i2,i2,i3,3x,i4,i3,2x,i7,i5,1x,i4,1x,a10)

* Initialize LU code counter
        jcount=1
* Compare LU for each polygon to each LU code and exit loop
* after match is found
        do 111 jjj=1,dim3
          if (cfname(jcount).eq.alandsoil) goto 1006
          jcount=jcount+1
111      continue
1006     continue
```

```
* Consider xeriscape for LSC
      if (loadcf(jcount,8).eq.2.and.xeriscape.eq.1) then
        frac=0.95
      else
        frac=1.0
      endif

* After match is found, increment acreage for corresponding use type (if use type > 0)

      if(loadcf(jcount,8).gt.0) then
        acres(row,col,loadcf(jcount,8))=
     +   min(acres(row,col,loadcf(jcount,8))+
     +   frac*polyarea*loadet(jcount,4)/43560.,2560.)
      endif

* Check if landuse code has 'R' = Wastewater REUSE polygon
      iflag=0
      do 119 jl=1,9
        code_up=ichar('R')
        code_low=ichar('R')+32
        if(ichar(alandsoil(jl:jl)).eq.code_low.or.
     +   ichar(alandsoil(jl:jl)).eq.code_up)then
          iflag=1
          goto 120
        endif
119     continue
120     continue


* Initialize flipoly to 0.0 (may be overwritten later on).
      flipoly=0.0
* If landuse code has 'R' (reuse;iflag=1) check if poly has LSC or GLF usetype and sum
* irrigated area for cell.
      if(iflag.eq.1) then
* Check if poly has LSC usetype
        if(loadcf(jcount,8).eq.2) then
* If polygon has no permit number, then polygon is definitely under reuse
          if(pwsprm.eq.'    ') then
            lan_reuse(row,col)=min(lan_reuse(row,col)+
     +       frac*polyarea*loadet(jcount,4)/43560.,2560.)
* Give error message and finish execution if poly has PWS permit # and is under reuse.
          else
            write(*,*) 'Check line ',k+1, ' in ',soilfile,
     +       'Polygon has R (reuse) code and PWS permit number'
            goto 121
          endif
* Check if poly has GLF use type
```

```
          else if (loadcf(jcount,8).eq.6) then
* If polygon has no permit number, then polygon is definitely under reuse
            if(pwsprm.eq.'     ') then
              glf_reuse(row,col)=min(glf_reuse(row,col)+
     +          frac*polyarea*loadet(jcount,4)/43560.,2560.)
* Give error message and finish execution if poly has PWS permit # and is under reuse.
            else
              write(*,*) 'Check line ',k+1, ' in ',soilfile,
     +          'Polygon has R (reuse) code and PWS permit number'
              goto 121
            endif
          endif
* If landuse does not have 'R' (iflag=0) do various checks to determine if polygon is
* irrigated from PWS
        else
* If east of saltwater interface --> ALL landscape irrigation is from PWS
          if(salt.eq.1) then
            flipoly=1.0
          else
* If no permit --> no landscape irrigation from PWS
            if(pwsprm .eq. '     ') then
              flipoly = 0.0
* If permit --> check LU code to assign fraction of polygon irrigated from PWS

            else
* Compare LU for each polygon to each LU code and exit loop
* after match is found
              do 127 jjl=1,dim3
                if (lu(jjl).eq.alandsoil) then
                  flipoly=flilu(jjl)
                  goto 1011
                endif
127             continue
1011             continue
            endif
          endif
c If polygon is of LSC usetype then sum irrigated area per cell
          if(loadcf(jcount,8).eq.2) then
            lan_pws(row,col)=min(lan_pws(row,col)+
     +        frac*flipoly*polyarea*loadet(jcount,4)/43560.,2560.)
          endif
        endif
109   continue
110   continue
      close(30)

* Make acreages for non-LEC cells equal to 0
```

```
      do 112 i=1,nrow
        do 113 j=1,ncol
          do 114 il=1,61
            if(i.eq.rowlec(il).and.j.ge.minco(il).and.
     +         j.le.maxco(il))goto 115
114         continue
          do 116 jj=1,ncrop
            acres(i,j,jj)=0.0
116         continue
115       continue

*     Calculate fraction of landscape irrigation from PWS (usetype 2 is LSC)
*     Landscape areas with reuse and PWS source will be part of acres(i,j,2)
          if (acres(i,j,2).gt.0.0) then
            fli(i,j) = lan_pws(i,j)/(acres(i,j,2))
          else
            fli(i,j) = 0.0
          endif

*     Calculate fraction of golf course irrigation from reuse (usetype 6 is GLF)
*     Golf course areas with reuse will be part of acres(i,j,6)
          if (acres(i,j,6).gt.0.0) then !use type 6 is golf course irrigation
            fgi(i,j) = glf_reuse(i,j)/(acres(i,j,6))
          else
            fgi(i,j) = 0.0
          endif

*     Calculate fraction of landscape irrigation from reuse (usetype 2 is LSC)
*     Landscape areas with reuse and PWS source will be part of acres(i,j,2)
          if (acres(i,j,2).gt.0.0) then
            flr(i,j) = lan_reuse(i,j)/(acres(i,j,2))
          else
            flr(i,j) = 0.0
          endif

113     continue
112   continue

* Open file where statdta will be written to
      open(60,file='statdta.dat',status='unknown')
* Write acreages for each cell and use type to statdta
      do 117 i=1,ncrop
        do 118 j=1,184
          ist = statc2(j)-statc1(j)+1
          write(60,1007) cuttype(ord(i)),statr(j),statc1(j),
     +      statc2(j),(acres(statr(j),icols,ord(i)),
     +      icols=statc1(j),statc2(j))
```

```
1007  format(a3,1x,i2,1x,i2,1x,i2,15f6.0)
118     continue
117   continue

* Write fraction of landscape irrigation from PWS
      do 122 j=1,184
        ist = statc2(j)-statc1(j)+1
        write(60,1008) 'FLI',statr(j),statc1(j),statc2(j),
     +   (fli(statr(j),icols),icols=statc1(j),statc2(j))
1008  format(a3,1x,i2,1x,i2,1x,i2,15f6.2)
122     continue

* Write fraction of golf course irrigation from reuse
      do 123 j=1,184
        ist = statc2(j)-statc1(j)+1
        write(60,1008) 'FGI',statr(j),statc1(j),statc2(j),
     +   (fgi(statr(j),icols),icols=statc1(j),statc2(j))
123     continue

* Write fraction of landscape irrigation from reuse
      do 124 j=1,184
        ist = statc2(j)-statc1(j)+1
        write(60,1008) 'FLR',statr(j),statc1(j),statc2(j),
     +   (flr(statr(j),icols),icols=statc1(j),statc2(j))
124     continue
121   continue

      end
```

Appendix C. Lookup table flipoly.dat used in create_statdta_all.f (Appendix B).

LU	flipoly
120	0.6
121	0.6
130	0.6
131	0.6
135	0.6
133	0.7
134	0.7
140	0.8
141	0.8
1411	0.8
142	0.8
143	0.8
144	0.8
145	0.8
147	0.8
148	0.8
170	0.8
171	0.8
172	0.8
173	0.8
174	0.8
175	0.8
176	0.8
180	0.8
183	0.8
185	0.8
186	0.8
187	0.8
188	0.8
189	0.8
253	0.8
811	0.8
813	0.8
833	0.8